
 
 
General rights 
Copyright and moral rights for the publications made accessible in the public portal are retained by the authors and/or other copyright 
owners and it is a condition of accessing publications that users recognise and abide by the legal requirements associated with these rights. 
 

 Users may download and print one copy of any publication from the public portal for the purpose of private study or research. 

 You may not further distribute the material or use it for any profit-making activity or commercial gain 

 You may freely distribute the URL identifying the publication in the public portal 
 
If you believe that this document breaches copyright please contact us providing details, and we will remove access to the work immediately 
and investigate your claim. 
  
 

   

 

 

Downloaded from orbit.dtu.dk on: May 26, 2023

vZ - An Optimizing SMT Solver

Bjørner, Nikolaj; Dung, Phan Anh; Fleckenstein, Lars

Published in:
Proceedings of the 21st International Conference on Tools and Algorithms for the Construction and Analysis of
Systems, TACAS 2015

Link to article, DOI:
10.1007/978-3-662-46681-0_14

Publication date:
2015

Link back to DTU Orbit

Citation (APA):
Bjørner, N., Dung, P. A., & Fleckenstein, L. (2015). vZ - An Optimizing SMT Solver. In Proceedings of the 21st
International Conference on Tools and Algorithms for the Construction and Analysis of Systems, TACAS 2015
(pp. 194-199). Springer. Lecture Notes in Computer Science Vol. 9035 https://doi.org/10.1007/978-3-662-46681-
0_14

https://doi.org/10.1007/978-3-662-46681-0_14
https://orbit.dtu.dk/en/publications/9694efc4-0f4d-4859-9e8f-44c90d314f3f
https://doi.org/10.1007/978-3-662-46681-0_14
https://doi.org/10.1007/978-3-662-46681-0_14


νZ - An Optimizing SMT Solver

Nikolaj Bjørner1, Anh-Dung Phan2, and Lars Fleckenstein3

1 Microsoft Research, Redmond, WA, USA
2 DTU Compute, Technical University of Denmark

3 Microsoft Dynamics
nbjorner@microsoft.com, padu@dtu.dk, LarsFleckenstein@outlook.com

Abstract. νZ is a part of the SMT solver Z3. It allows users to pose
and solve optimization problems modulo theories. Many SMT applica-
tions use models to provide satisfying assignments, and a growing num-
ber of these build on top of Z3 to get optimal assignments with respect
to objective functions. νZ provides a portfolio of approaches for solving
linear optimization problems over SMT formulas, MaxSMT, and their
combinations. Objective functions are combined as either Pareto fronts,
lexicographically, or each objective is optimized independently. We de-
scribe usage scenarios of νZ, outline the tool architecture that allows
dispatching problems to special purpose solvers, and examine use cases.

1 An Invitation to νZ

νZ extends the functionality of Z3 [7] to include optimization objectives. It allows
users to solve SMT constraints and at the same time formulate optimality criteria
for the solutions. It relieves users of Z3 from writing their own loops around the
solver to find optimal values. The solver integrates state-of-the-art algorithms
for optimization, and it extends some of these algorithms with its own twists: For
example, it includes direct support for difference logic solvers, it uses Simplex
over non-standard numbers to find unbounded constraints, and it applies an
incremental version of the MaxRes [11] algorithm for MaxSAT solving.

(declare-fun x () Int)

(declare-fun y () Int)

(assert (and (< y 5) (< x 2)))

(assert (< (- y x) 1))

(maximize (+ x y))

(check-sat)

(get-model)

To give a first idea, we can ask to
optimize the term x+y under the con-
straints y < 5 ∧ x < 2 and y − x < 1
using the SMT query to the right.
The optimal answer is given as 2 and
νZ returns a model where x = y =
1. The example shows the maximize

command that is added to the SMT-
LIB [13] syntax.

1.1 Optimization Commands

The full set of commands νZ adds to SMT-LIB are:



(declare-fun x () Int)

(declare-fun y () Int)

(define-fun a1 () Bool (> x 0))

(define-fun a2 () Bool (< x y))

(assert (=> a2 a1))

(assert-soft a2 :dweight 3.1)

(assert-soft (not a1) :weight 5)

(check-sat)

(get-model)

Fig. 1. Maximize 3.1 ·a2 + 5 ·a1. νZ finds
a solution where y ≤ x ≤ 0.

(declare-fun x () Int)

(declare-fun y () Int)

(assert (= (+ x y) 10))

(assert (>= x 0))

(assert (>= y 0))

(maximize x)

(maximize y)

(set-option :opt.priority box)

(check-sat)

Fig. 2. νZ produces two independent op-
tima x = 10, respectively y = 10.

– (maximize t ) - instruct the solver to maximize t . The type of the term t

can be either Integer, Real or Bit-vector.

– (minimize t ) - instruct the solver to minimize t .

– (assert-soft F [:weight n | :dweight d ] [:id id ]) - assert soft con-
straint F , optionally with an integral weight n or a decimal weight d . If no
weight is given, the default weight is 1 (1.0). Decimal and integral weights
can be mixed freely. Soft constraints can be furthermore tagged with an
optional name id. This enables combining multiple different soft objectives.
Fig. 1 illustrates a use with soft constraints.

1.2 Combining Objectives

Multiple objectives can be combined using lexicographic, Pareto fronts or as
independent box objectives.

Lexicographic combinations: By default, νZ maximizes objectives t1, t2 sub-
ject to the constraint F using a lexicographic combination. It finds a model M ,
such that M satisfies F and the pair 〈M(t1),M(t2)〉 is lexicographically maxi-
mal. In other words, there is no model M ′ of F , such that either M ′(t1) > M(t1)
or M ′(t1) = M(t1), M ′(t2) > M(t2).

Pareto fronts: Again, given two maximization objectives t1, t2, the set of Pareto
fronts under F are the set of models M1, . . . ,Mi, . . . ,Mj , . . ., such that either
Mi(t1) > Mj(t1) or Mi(t2) > Mj(t2), and at the same time either Mi(t1) <
Mj(t1) or Mi(t2) < Mj(t2); and for each Mi, there is no M ′ that dominates Mi.
νZ uses the Guided Improvement Algorithm [14] to produce multiple objectives.
Fig. 3 illustrates a use where Pareto combination is specified.

Boxes: Box objectives, illustrated in Fig.2 are used to specify independent op-
tima subject to a formula F . They are used in the Symba tool [9]. The box
combination of objectives t1, t2 requires up to two models M1,M2 of F , such
that M1(t1) is the maximal value of t1 and M2(t2) is the maximal value for t2.



1.3 Programming Optimization

The optimization features are available over Z3’s programmatic APIs for C,
C++, Java, .NET, and Python. There is furthermore a library available as an
example that plugs into the Microsoft Solver Foundation (MSF). Fig. 3 shows
an example using the Python API to generate Pareto optimal solutions. Fig. 4
shows an OML model used by MSF.

x, y = Ints(’x y’)

opt = Optimize()

opt.set(priority=’pareto’)

opt.add(x + y == 10, x >= 0, y >= 0)

mx = opt.maximize(x)

my = opt.maximize(y)

while opt.check() == sat:

print mx.value(), my.value()

Fig. 3. Pareto optimization in Python. νZ
produces all 11 Pareto fronts.

Model[

Decisions[

Reals[-Infinity, Infinity], xs, xl ],

Constraints[

limits -> 0 <= xs & 0 <= xl,

BoxWood -> xs + 3 * xl <= 200,

Lathe -> 3 * xs + 2 * xl <= 160 ],

Goals[

Maximize[ $ -> 5 * xs + 20 * xl ]]]

Fig. 4. OML model used by MSF.

1.4 MILP, MaxSAT, CP and SMT

Efficient mixed integer linear solvers are backbones of several highly tuned tools,
such as CPLEX and Gurobi, used in operations research contexts. Being able to
state and solve optimization objectives in the context of logical constraints has
also been well recognized in the SMT community [12,5,15,8] and it is a recurring
feature request for Z3 as well. We briefly outline a use case in Section 4, and
through this experience we observed a need for more abstract and flexible ways of
modeling problems than exposed by OML used by the Microsoft Solver Founda-
tion (MSF), where flexible Boolean combinations of constraints, which empower
end-users to refine models, are afterthoughts. By making νZ generally available,
we hope to make it easier for existing users to use Z3, for instance [2], and to fuel
further applications that benefit from the flexibility and expressive power of Z3’s
SMT engines, including theory support and quantifiers, with the convenience of
built-in support for (reasonably tuned) optimization algorithms. In return, we
anticipate that new applications from SMT users can inspire advances in ar-
eas such as non-linear arithmetic, mixed symbolic/numerical algorithms, and
combinations with Horn clauses.

1.5 Resources

The full source code of νZ is available with Z3 from http://z3.codeplex.com,
the sources compile on all main platforms, there is an online tutorial on http://

rise4fun.com/z3opt/tutorial/, and a companion paper [3] describes details
of algorithms used in νZ.

http://z3.codeplex.com
http://rise4fun.com/z3opt/tutorial/
http://rise4fun.com/z3opt/tutorial/


2 Architecture

Fig. 5 gives an architectural overview of νZ. The input SMT formulas and ob-
jectives are rewritten and simplified using a custom strategy that detects 0-1
integer variables and rewrites these into Pseudo-Boolean Optimization (PBO)
constraints. Objective functions over 0-1 variables are rewritten as MaxSAT
problems4. If there are multiple objectives, then νZ orchestrates calls into the
SMT or SAT cores. For box constraints over reals, νZ combines all linear arith-
metic objectives and invokes a single instance of the OptSMT engine; for lexico-
graphic combinations of soft constraints, νZ invokes the MaxSAT engine using
multiple calls.

0-1 constraints
⇒ PBO

SMT formula
with objectives

Combination of
objective functions

OptSMT: Arithmetic MaxSMT: Soft Constraints

PB and
Cost solvers

SMT solver SAT solver

Fig. 5. νZ system architecture

3 Internals

OptSMT: We have augmented Z3’s dual Simplex core with a primal phase
that finds maximal assignments for reals. It also improves bounds on integers as
long as the improvements are integral. It is used, similarly to [15,9], to improve
values of objective functions. A similar primal Simplex solver is also accessible
to Z3’s difference logic engines. νZ discovers unbounded objectives by using
non-standard arithmetic: It checks if t ≥ ∞ is feasible, over the extension field
R∪{ε,∞ := 1/ε}. This contrasts the approach proposed in [9] that uses a search
through hyper-planes extracted from inequalities.

νZ also contains a Pseudo-Boolean theory solver. It borrows from [4,1] for
simplification, generating conflict clauses, and incrementally compiling into small
sorting circuits. It also adds an option to prune branches using dual simplex.
MaxSMT: νZ implements several engines for MaxSAT. These include WMax [12],
MaxRes [11], BCD2 [10], MaxHS [6]. WMax uses a specialized theory solver of

4 using the correspondence: maximize c1 · x1 + c2 · x2 ≡ (assert-soft x1 :weight c1),
(assert-soft x2 :weight c2)



costs, also explored in [5]. The solver associates penalties with a set of tracked
propositional variables. It then monitors the truth assignments to these variables,
as given by the SAT solver. The cost is incremented when a tracked variable is
assigned to false. The solver creates a conflict clause when the cost exceeds the
current optimal value. WMax can be interrupted at any point with a current
upper bound. Our implementation of MaxRes generally performs much better
than WMax. MaxRes increments a lower bound when there is an unsatisfiable
core of the soft constraints. It then replaces the core F1, . . . , Fk with new soft
constraints F ′1, F

′
2, . . . , F

′
k−1 using the equations:

F ′1 = F2 ∨ F1, F
′
2 = F3 ∨ (F1 ∧ F2), . . . , F ′k−1 = Fk ∨ ((F1 ∧ F2) ∧ . . . ∧ Fk−1) .

SAT: νZ reduces Pseudo-Boolean formulas to propositional SAT by converting
cardinality constraints using sorting circuits, using a Shannon decomposition
(BDDs) of simple PB inequalities and falling back to bit-vector constraints on
inequalities where the BDD conversion is too expensive. This transformation is
available by ensuring that the option :opt.enable sat is true. For benchmarks
that can be fully reduced to propositional SAT, MaxRes uses Z3’s SAT solver.

4 A use for νZ

As a driving scenario for νZ we used an experimental warehouse manager in the
context of Microsoft Dynamics AX. The objective is to reduce cost by optimizing
how shipments are distributed on trucks, reducing the number of trucks, the
distance traveled by the truck while maximizing the amount of goods delivered.
AX can deliver the standard constraints and cost functions, e.g., weight and
volume of a truck, but users often want to be more specific. For example, frozen
foods need to be in a cooled truck and cannot be packed together with chemicals.
The expressive power and convenience of SMT is useful: these constraints can
be formulated as a Boolean combination of linear constraints over 0-1 variables,
while the objective functions we considered could be expressed as lexicographic
combinations of a couple of cost functions. Such cost functions are expected to
evolve when users learn more about their usages. The abstraction layer of the
models provides this flexibility.

4.1 Experience

We evaluated νZ on a cross-section of benchmarks used in MaxSAT competi-
tions, from Z3 users, and from recent publications. Table 1 summarizes a se-
lected evaluation. Motivating examples from users included strategy scheduling
for Vampire (MaxSAT) that are easy with the new MaxSAT engine, but used to
be hard for the bisection search used by Vampire. Likewise, Cezary Kaliszyk has
used Z3 to tune his portfolio solver using linear arithmetic constraints. His sys-
tems are significantly more challenging (take days to run). In this case WMax
offers partial solutions during search. Elvira Albert tried using Z3 for finding



longest paths, her benchmarks are called bb (≈300 clauses), chat (≈3K clauses)
and p2p (≈30K clauses), and we summarize timing for bb and chat below. The
p2p category remains really hard. Korovin and White experimented with en-
coding RNA alignment into MaxSAT, and Pandya and Bresler encoded DNA
alignment problems into MaxSMT over LIA. Many problems remain challenging,
including material discovery benchmarks from Ronan Le Bras, Pareto problems
from Oded Maler and even benchmarks from the pure Pseudo-Boolean domain 5.

Source Category Solved instances Time

MaxSAT 2014 wpms industrial track MaxSAT 361/410 0.5-1800s

MaxSAT 2014 pms industrial track MaxSAT 406/568 0.5-1800s

Longest Paths MaxSAT bb 8/8 <0.05s

Longest Paths MaxSAT chat 34/34 1-36s

DAL Allocation challenge PBO SampleA&B 96/96 0.02-6s

Symba [9] LRA 2435/2435 0.2s-36s

OptiMathSAT [15] LRA 9 non-random 0.5-20s
Table 1. Evaluation of νZ on selected examples

Non-termination Using Max-SMT. In CAV, pages 779–796, 2014.
9. Y. Li, A. Albarghouthi, Z. Kincaid, A. Gurfinkel, and M. Chechik. Symbolic

optimization with SMT solvers. In POPL, pages 607–618, 2014.
10. A. Morgado, F. Heras, and J. Marques-Silva. Improvements to Core-Guided Binary

Search for MaxSAT. In SAT, pages 284–297, 2012.
11. N. Narodytska and F. Bacchus. Maximum Satisfiability Using Core-Guided

MaxSAT Resolution. In AAAI, pages 2717–2723, 2014.
12. R. Nieuwenhuis and A. Oliveras. On SAT Modulo Theories and Optimization

Problems. In SAT, pages 156–169, 2006.
13. S. Ranise and C. Tinelli. The SMT Library (SMT-LIB). www.SMT-LIB.org, 2006.
14. D. Rayside, H.-C. Estler, and D. Jackson. The Guided Improvement Algorithm.

Technical Report MIT-CSAIL-TR-2009-033, MIT, 2009.
15. R. Sebastiani and S. Tomasi. Optimization in SMT with LA(Q) Cost Functions.

In IJCAR, pages 484–498, 2012.

5 http://www.cril.univ-artois.fr/PB12/

http://maxsat.ia.udl.cat/benchmarks/
http://maxsat.ia.udl.cat/benchmarks/
http://www.lifl.fr/LION9/challenge.php
http://www.cril.univ-artois.fr/PB12/


A Demonstration

We take a quick tour on available features in νZ. This section is based on the
online tutorial at http://rise4fun.com/Z3Opt/tutorial/guide. Please visit
the online playground for a quick try at νZ. More details of internals in νZ are
also available in a paper [3] that accompanies an invited talk at SCSS 2014. The
paper describes, at a high level, the algorithms used for different optimization
strategies in νZ.

Linear Arithmetic

Using the example in Fig. 2, νZ returns two independent values 10 and 10 for
two objectives x and y.

This example can be reproduced at http://rise4fun.com/Z3Opt/o7h.

Soft Constraints

The example in Fig. 1 demonstrates soft constraints with integral and decimal
weights. νZ gives an optimal model where the total weight equals to 3.1 and
x = y = 0.

http://rise4fun.com/Z3Opt/tutorial/guide
http://rise4fun.com/Z3Opt
http://rise4fun.com/Z3Opt/o7h


This example can be reproduced at http://rise4fun.com/Z3Opt/GR.

Programmatic APIs

νZ’s APIs are available for multiple languages. This section presents program-
matic APIs using a few F# code snippets.

Optimization features are facilitated through the new Optimize solver. As
being shown in Fig. 6, combining objectives can be set via solver parameters.
MkMinimize and Value are used to set optimization goals and retrieve optimal
values respectively (see Fig. 7).

PB constraints are available via MkAtMost and MkPBLe commands (see
Fig. 8). They allow users to express some problems in a more concise man-
ner. Soft constraints can be sent to the solver using the AssertSoft command.
As Fig. 7 demonstrated, users can employ Check and Value methods for getting
optimal values.

http://rise4fun.com/Z3Opt/GR


use context = new Context()

let solver = context.MkOptimize()

let param = context.MkParams()

param.Add("priority", "pareto")

solver.Parameters <- param

Fig. 6. Combining objectives in F#.

let cost =

ys

|> Array.map2 ( *. ) costs

|> add

|> solver.MkMinimize

match solver.Check() with

| Status.SATISFIABLE ->

Some cost.Value

| _ -> None

Fig. 7. Getting minimized cost in F#.

let mkAtmost k bs =

context.MkAtMost(bs, uint32 k)

for xi in xs do

solver.Assert (mkAtmost 1 xi)

Fig. 8. Creating PB constraints in F#.

let serverNum =

seq {

for j in 1..n ->

let yj = context.MkNot(ys.[j-1])

server.AssertSoft(yj, 1u,

"num_servers")

} |> Seq.last

Fig. 9. Asserting soft constraints in F#.

The code fragments above are parts of a complete case study. More details
can be found in the introductory blog post at http://lonelypad.blogspot.

dk/2014/08/f-and-linear-programming-introduction.html.

Microsoft Solver Foundation plugin

SolverContext context = SolverContext.GetContext();

Model model = context.CreateModel();

Decision x1 = new Decision(Domain.RealRange(0, 2), "x1");

Decision x2 = new Decision(Domain.RealRange(0, 2), "x2");

Decision z = new Decision(Domain.IntegerRange(0, 1), "z");

model.AddDecisions(x1, x2, z);

model.AddConstraint("Row0", x1 - z <= 1);

model.AddConstraint("Row1", x2 + z <= 2);

Goal goal = model.AddGoal("Goal0", GoalKind.Maximize, x1 + x2);

Solution solution = context.Solve(new Z3MILPDirective());

Fig. 10. Using νZ MSF plugin in C#.

νZ is also available as a plugin to MSF framework. It allows users to take
advantage of MSF’s comprehensive infrastructure. For example, users are able

http://lonelypad.blogspot.dk/2014/08/f-and-linear-programming-introduction.html
http://lonelypad.blogspot.dk/2014/08/f-and-linear-programming-introduction.html


to run multiple solvers in parallel with νZ inside MSF. The plugin plays a vital
role in bootstrapping νZ by comparing it with other solvers in MSF.

Case study

In this section, we use the scenario of a warehouse manager to demonstrate
how νZ can be used for solving problems inspired by Dynamics AX scenarios.
Suppose that there are a number of goods that need to be shipped to customers.
These goods have specific destinations and requirements as shown in Table 2.

Shipment no. Weight(kg) Volume (m3) Requirement Zip code

1 400 300 Dry 2112

2 300 350 Fresh 2100

3 220 160 Dry 2103
Table 2. Shipments that requires transportation.

Table 3 describes trucks along with their capacities, tentative destinations
and associated costs. The constraints are:

1. Shipments have to fit in the capacities given by the trucks.
2. Product requirements have to be fulfilled.
3. Shipments are delivered to correct addresses (zip codes).

We can state the goals in their order of priorities (1) Use as few trucks as possible
(2) Minimize the total cost for transportation. The scenario is a variant of a bin
packing problem. See also the ESICUP Challenge.

Truck no. Weight(kg) Volume (m3) Requirement Zip codes Initial cost (USD)

1 777 700 Fresh 2100, 2103 100

2 450 1000 Dry 2100, 2103, 2112 120

3 600 460 Dry 2100, 2112 130
Table 3. Available trucks for transportation.

The constraints and goals can be modeled directly as 0-1 variables. Let the
0-1 variable xij denote that shipment i is packed into truck j. Each shipment
has to be in exactly one truck.

Let 0-1 variable yj denote that truck j is in use. Trucks are used if something
is packed into them:

Here we describe weight as examples of capacity constraints. There are similar
constraints for volumes.

To keep this example simple we here treat zip codes as fixed costs. The goals
can be modeled as in Fig. 14. Note that we would like to minimize the number

http://challenge-esicup-2015.org/


(assert (= (+ x_1_1 x_1_2 x_1_3) 1))

(assert (= (+ x_2_1 x_2_2 x_2_3) 1))

(assert (= (+ x_3_1 x_3_2 x_3_3) 1))

Fig. 11. Each shipment is in only one truck.

(define-fun imax ((a Int) (b Int)) Int (if (> a b) a b))

(assert (= y_1 (imax (imax x_1_1 x_2_1) x_3_1)))

(assert (= y_2 (imax (imax x_1_2 x_2_2) x_3_2)))

(assert (= y_3 (imax (imax x_1_3 x_2_3) x_3_3)))

Fig. 12. Truck j (yj) is in use.

of used trucks and minimize total cost, so in MaxSMT these goals are equivalent
to maximizing number of unused trucks and maximizing saving.

Although we express the problem via 0-1 variables, νZ translates it via
Pseudo-Boolean constraints (opt.elim 01=true by default). In this case νZ
uses the Pseudo-Boolean solver and solves a MaxSMT problem. Furthermore,
by ensuring that opt.enable sat=true, opt.maxsat engine=maxres, νZ
translates Pseudo-Boolean constraints to SAT and uses a pure MaxSAT solver.

The example illustrates a fair amount of flexibility in formulating constraints,
and control over the most suitable back-end for these constraints.

Selected evaluation details

In Fig. 15, the two top graphs show comparison with νZ and the best solvers
in MaxSAT 2014 competition. Note that νZ is presently not fine-tuned as some
of these competition entrants are, but performs quite respectably. What is not
shown is performance of other solvers entering the competition and performing
on par or worse than νZ. The bottom graph shows relative comparison of νZ
with and without the Symba approach of finding unbounded objectives. The
experience indicates that searching separately for unbounded objectives does
not really help.

(assert (<= (+ (* 400 x_1_1) (* 300 x_2_1) (* 220 x_3_1)) (* 777 y_1)))

(assert (<= (+ (* 400 x_1_2) (* 300 x_2_2) (* 220 x_3_2)) (* 450 y_2)))

(assert (<= (+ (* 400 x_1_3) (* 300 x_2_3) (* 220 x_3_3)) (* 600 y_3)))

Fig. 13. Capacity constraints.



(assert-soft (= y_1 0) :id unused_trucks)

(assert-soft (= y_2 0) :id unused_trucks)

(assert-soft (= y_3 0) :id unused_trucks)

(minimize (+ (* 100 y_1) (* 20 (+ x_1_1 x_2_1 x_3_1))

(* 120 y_2) (* 30 (+ x_1_2 x_2_2 x_3_2))

(* 130 y_3) (* 10 (+ x_1_3 x_2_3 x_3_3))))

(set-option :opt.priority lex)

(check-sat)

Fig. 14. Two optimization goals of the truck planning example.

Fig. 15. Evaluation results.

